

Περιγραφή

Η συνάρτηση **COUNTIF** καταμετρά τους αριθμούς των κελιών σε μια περιοχή που ικανοποιεί ένα κριτήριο που καθορίζετε. Για παράδειγμα, μπορείτε να καταμετρήσετε όλα τα κελιά που ξεκινούν με ένα συγκεκριμένο γράμμα ή όλα τα κελιά που περιέχουν έναν αριθμό που είναι μεγαλύτερος ή μικρότερος από τον αριθμό που καθορίζετε. Για παράδειγμα, ας υποθέσουμε ότι έχετε ένα φύλλο εργασίας που περιέχει μια λίστα εργασιών στη στήλη Α και το όνομα του ατόμου στο οποίο έχει ανατεθεί κάθε εργασία στη στήλη Β. Μπορείτε να χρησιμοποιήσετε τη συνάρτηση **COUNTIF** για να καταμετρήσετε πόσες φορές το όνομα ενός ατόμου θα εμφανίζεται στη στήλη Β και με αυτόν τον τρόπο να καθορίσετε πόσες εργασίες έχουν ανατεθεί στο συγκεκριμένο άτομο. Για παράδειγμα:

=COUNTIF(B2:B25;"Στέλλα")

Σημείωση Για την καταμέτρηση κελιών με βάση πολλαπλά κριτήρια, ανατρέξτε στο θέμα [Συνάρτηση COUNTIFS](#).

Σύνταξη

COUNTIF(περιοχή; κριτήρια)

Η σύνταξη της συνάρτησης COUNTIF περιλαμβάνει τα παρακάτω όρισμα:

- **περιοχή** Υποχρεωτικό. Πρόκειται για ένα ή περισσότερα κελιά που θα καταμετρηθούν, συμπεριλαμβανομένων ονομάτων, πινάκων ή αναφορών που περιέχουν αριθμούς. Οι τιμές που είναι κενές ή κείμενο παραβλέπονται.
- **κριτήρια** Υποχρεωτικό. Πρόκειται για αριθμό, παράσταση, αναφορά κελιού ή συμβολοσειρά κειμένου που καθορίζει ποια κελιά θα καταμετρηθούν. Για παράδειγμα τα κριτήρια μπορούν να εκφραστούν ως 32, ">32", B4, "μήλα" ή "32".

Σημείωση

- Μπορείτε να χρησιμοποιήσετε τους χαρακτήρες μπαλαντέρ — το ερωτηματικό (?) και τον αστερίσκο (*) — στα κριτήρια. Το ερωτηματικό αντιστοιχεί σε έναν οποιονδήποτε χαρακτήρα και ο αστερίσκος αντιστοιχεί σε μία οποιαδήποτε ακολουθία χαρακτήρων. Εάν θέλετε να εντοπίσετε ένα πραγματικό ερωτηματικό ή αστερίσκο, πληκτρολογήστε το σύμβολο περισπωμένης (~) πριν από το χαρακτήρα.
- Στα κριτήρια δεν γίνεται διάκριση πεζών-κεφαλαίων. Για παράδειγμα, η συμβολοσειρά "πριν" και η συμβολοσειρά "ΠΡΙΝ" θα οδηγήσουν στην εμφάνιση των ίδιων κελιών.

Παρατηρήσεις

- Η συνάρτηση COUNTIF επιστρέφει την τιμή σφάλματος #ΤΙΜΗ!, όταν την χρησιμοποιείτε για να αντιστοιχίσετε συμβολοσειρές μεγαλύτερες από 255 χαρακτήρες.

Παραδείγματα

Στο βιβλίο εργασίας που ακολουθεί παρουσιάζονται παραδείγματα αυτής της συνάρτησης. Για να μάθετε πώς λειτουργεί η συνάρτηση, δείτε τα παραδείγματα, αλλάξτε τους υπάρχοντες τύπους ή εισαγάγετε τους δικούς σας.

Αντιγράψτε τα δεδομένα του παραδείγματος στον πίνακα που ακολουθεί και, στη συνέχεια, επικολλήστε τα στο κελί A1 ενός νέου φύλλου εργασίας του Excel. Για να εμφανιστούν τα αποτελέσματα των τύπων, επιλέξτε τους, πατήστε το πλήκτρο F2 και, στη συνέχεια, πατήστε το πλήκτρο Enter. Εάν χρειάζεται, μπορείτε να ρυθμίσετε το πλάτος των στηλών για να βλέπετε όλα τα δεδομένα.

Δεδομένα	Δεδομένα	
μήλα	32	
πορτοκάλια	54	
ροδάκινα	75	
μήλα	86	
Τύπος	Περιγραφή	Αποτέλεσμα
=COUNTIF(A2:A5;"μήλα")	Αριθμός κελιών με μήλα στα κελιά A2 έως A5.	2
=COUNTIF(A2:A5;A4)	Αριθμός κελιών με ροδάκινα στα κελιά A2 έως A5.	1
=COUNTIF(A2:A5;A3)+COUNTIF(A2:A5;A2)	Αριθμός κελιών με πορτοκάλια και μήλα στα κελιά A2 έως A5.	3
=COUNTIF(B2:B5;"> 55")	Αριθμός κελιών με τιμή μεγαλύτερη από 55 στα κελιά B2 έως B5.	2
=COUNTIF(B2:B5;"<>"&B4)	Αριθμός κελιών με τιμή που δεν είναι ίση με 75 στα κελιά B2 έως B5.	3
=COUNTIF(B2:B5;">=32")-COUNTIF(B2:B5;">85")	Αριθμός κελιών με τιμή μεγαλύτερη ή ίση με 32 και μικρότερη ή ίση με 85 στα κελιά B2 έως B5.	3

Για να εργαστείτε πιο αναλυτικά με τα δεδομένα του παραδείγματος στο Excel, κάντε [λήψη του ενσωματωμένου βιβλίου εργασίας](#) στον υπολογιστή σας και, στη συνέχεια, ανοίξτε το στο Excel.

Παράδειγμα 2

Δεδομένα

μήλα	Ναι
πορτοκάλια	ΟΧΙ
ροδάκινα	Όχι
μήλα	ναι

Τύπος	Περιγραφή	Αποτέλεσμα
=COUNTIF(A2:A7;"*λα")	Αριθμός κελιών που τελειώνουν με τα γράμματα "λα" στα κελιά A2 έως A7.	4
=COUNTIF(A2:A7;"??λα")	Αριθμός κελιών που τελειώνουν με τα γράμματα "λα" και έχουν ακριβώς 4 χαρακτήρες στα κελιά A2 έως A7.	2
=COUNTIF(A2:A7;"*")	Αριθμός κελιών που περιέχουν οποιοδήποτε κείμενο στα κελιά A2 έως A7.	4
=COUNTIF(A2:A7;"<>"&"*")	Αριθμός κελιών που δεν περιέχουν κείμενο στα κελιά A2 έως A7.	2
=COUNTIF(B2:B7;"Όχι") / ROWS(B2:B7)	Ο μέσος αριθμός αρνητικών ψήφων "Όχι" (συμπεριλαμβανομένων των κενών κελιών) στα κελιά B2 έως B7.	0,33333333
=COUNTIF(B2:B7;"Ναι") / (ROWS(B2:B7) -COUNTIF(B2:B7;"<>"&"*"))	Ο μέσος αριθμός θετικών ψήφων "Ναι" (εξαιρουμένων των κενών κελιών) στα κελιά B2 έως B7.	0,5